# Maintainer: Antonio Rojas <arojas@archlinux.org>

pkgbase=cysignals
pkgname=python-cysignals
pkgver=1.11.2
pkgrel=3
pkgdesc='Interrupt and signal handling for Cython'
arch=(loong64 x86_64)
url='https://www.sagemath.org'
license=(LGPL)
depends=(python pari)
makedepends=(cython python-setuptools)
conflicts=(python2-cysignals python2-fpylll python2-cypari2 python2-pplpy)
source=(https://pypi.python.org/packages/source/c/cysignals/cysignals-$pkgver.tar.gz)
sha256sums=('5858b1760fbe21848121b826b2463a67ac5a45caf3d73105497a68618c5a6fa6')

build() {
  cd $pkgbase-$pkgver
  CFLAGS+=" -Wp,-U_FORTIFY_SOURCE" \
  python setup.py build
}

package_python-cysignals() {
  cd $pkgbase-$pkgver
  python setup.py install --root="$pkgdir"
}
